3　数

Mathematica では，数あるいは数値はどのように扱われるかを説明する．

■ 数の型

Mathematica で扱える数には，整数，有理数，実数（浮動小数点数），複素数の４種類があり，それぞれInteger, Rational, Real, Complex という頭部で表されている．頭部については，通常は気にする必要はないが，プログラミングの際に，その知識が必要となる時がある．頭部が何であるかを調べるには，Headという関数を用いる．

■ 整数

..., -3, -2, -1, 0, 1, 2, 3, ...が整数である． Mathematica では，整数には Integer という頭部がついている．

```
In[1]:= Head[-3]

Out[1]= Integer
```

Mathematica は任意桁の整数を扱える． 100 の階乗を求めてみる．

```
In[2]:= 100!

Out[2]= 93326215443944152681699238856266700490715968264381621468592963895217599993229915\
          608941463976156518286253697920827223758251185210916864000000000000000000000000
```

出力の右端の\は，結果が次の行に続くことを表す．

■ 有理数

Mathematica では，$2 \div 3$ のような，整数の割算をすると， 0.666667
などという近似値ではなく， $\frac{2}{3}$ という有理数を返す．有理数の頭部は**Rational**である.

```
In[3]:= 2 / 3
```

Out[3]= $\frac{2}{3}$

```
In[4]:= Head[%]

Out[4]= Rational
```

有理数の計算を行うと，分母を通分して，約分をしたものを返す．

```
In[5]:= 1 + 2 / 3 + 3 / 4
```

Out[5]= $\frac{29}{12}$

有理数を小数点を用いた近似値に変換するには N を用いる.

```
In[6]:=  N[1 + 2 / 3 + 3 / 4]

Out[6]=  2.41667
```

■ 実数（浮動小数点数）

小数点を含む数は実数（浮動小数点数）すなわち，ある精度をもった近似値として扱われる．実数の頭部はRealである.

```
In[7]:=  Head[12.3]

Out[7]=  Real
```

実数（浮動小数点数）と整数や有理数などとの演算結果は実数（浮動小数点数）である.

```
In[8]:=  12.2 * 5 + 4 / 7

Out[8]=  61.5714

In[9]:=  Head[%]

Out[9]=  Real
```

■ 複素数

複素数はComplexを頭部にもつ．また，複素数を表すための虚数単位 $i = \sqrt{-1}$ を入力するには，大文字の I を用いる．ただし， 画面上の出力では ⅈ という文字で表される.

```
In[10]:=  2 + 3 I

Out[10]=  2 + 3 ⅈ
```

複素数2+3Iの完全形式はPlus[2, Times[3, I]]ではなく，Complex[2, 3]である.

```
In[11]:=  FullForm[%]

Out[11]//FullForm=
       Complex[2, 3]
```

複素数の計算を行うと， Mathematica は分母の有理化を行い， a+b i という形にする.

```
In[12]:=  z = (3 + 7 I) / (1 + 2 I) - 3
```

Out[12]= $\frac{2}{5} + \frac{ⅈ}{5}$

```
In[13]:=  FullForm[z]

Out[13]//FullForm=
       Complex[Rational[2, 5], Rational[1, 5]]
```

複素数の実部，虚部，絶対値，偏角，共役複素数はそれぞれ，Re, Im, Abs, Arg, Conjugateという組み込み関数で求められる.

In[14]:= **Re[z]**

Out[14]= $\frac{2}{5}$

In[15]:= **Im[z]**

Out[15]= $\frac{1}{5}$

In[16]:= **Abs[z]**

Out[16]= $\frac{1}{\sqrt{5}}$

In[17]:= **Arg[z]**

Out[17]= $\mathrm{ArcTan}\left[\frac{1}{2}\right]$

In[18]:= **Conjugate[z]**

Out[18]= $\frac{2}{5}-\frac{i}{5}$

実部および虚部は浮動小数点数であってもよい.

In[19]:= **3.5 + 1.2 I**

Out[19]= 3.5 + 1.2 i

In[20]:= **FullForm[%]**

Out[20]//FullForm=
```
Complex[3.5`, 1.2`]
```

■ 無理数

Mathematicaで扱える数には整数，有理数，実数（浮動小数点数），複素数の４種類があるが，この他にも，数学的にきちんとした数値をもっているものもある．たとえば，$\sqrt{2}$ や π などの無理数である．Mathematica はこのような数値は近似値で表すことをしないで，厳密な表現のままで計算を行う．

In[21]:= **Sqrt[2] + Sqrt[3]**

Out[21]= $\sqrt{2}+\sqrt{3}$

これは $\sqrt{2}+\sqrt{3}$ 以外には厳密な表現方法はない．この値を二乗しても，それは単に二乗したものが返ってくるだけで，展開はされない．

In[22]:= **%^2**

Out[22]= $\left(\sqrt{2}+\sqrt{3}\right)^2$

展開を行うには，Expand という関数を用いる．

In[23]:= **Expand[%]**

Out[23]= $5+2\sqrt{6}$

平方根以外の無理数，たとえば円周率 π や自然対数の底eなども，厳密な形のまま計算を行う．

円周率 π は Pi というシンボルで入力されるが，表示は π となる．

```
In[24]:= Pi / 3
```

Out[24]= $\frac{\pi}{3}$

自然対数の底 E というシンボルで入力されるが，表示は e となる．

```
In[25]:= E^(2 / 3)
```

Out[25]= $e^{2/3}$

しかし， Mathematica はその意味を理解しているので，式が簡約化できるならば，簡単な式に直して返す．

```
In[26]:= Sin[Pi / 3]
```

Out[26]= $\frac{\sqrt{3}}{2}$

```
In[27]:= Log[E^(2 / 3)]
```

Out[27]= $\frac{2}{3}$

■ 近似値

N[expr, n]は，式 expr の数値を有効桁数 n 桁の近似値として表す． n を省略すると，機械精度（次で説明）の近似値を返す．

```
In[28]:= N[Pi, 100]

Out[28]= 3.1415926535897932384626433832795028841971693993751058209749445923078164062862089\
         98628034825342117068

In[29]:= N[Sin[Pi / 3]]

Out[29]= 0.866025
```

■ 数値の精度

Mathematica で扱う数値はすべて，ある精度をもっている．数値の精度すなわち有効桁数は， Precision という関数を使えば知ることができる．

■ 無限精度

整数，有理数や E,Pi などの数学定数は，厳密な値として，無限大の精度をもっている．

```
In[30]:= Precision[3]

Out[30]= ∞
```

```
In[31]:= Precision[2 / 3]

Out[31]= ∞

In[32]:= Precision[Pi]

Out[32]= ∞
```

これは，求めようと思えば，何桁までも正確な値を求めることが可能であることを意味している．

```
In[33]:= N[Pi, 300]

Out[33]= 3.14159265358979323846264338327950288419716939937510582097494459230781640628620899
         86280348253421170679821480865132823066470938446095505822317253594081284811174502
         84102701938521105559644622948954930381964428810975665933446128475648233786783165
         27120190914564856692346034861045432664821339360726024914127
```

■ 機械精度

小数点を含む実数（浮動小数点数）は有限の精度をもっている．通常，それは機械精度と呼ばれる精度である．

```
In[34]:= a = 3.14

Out[34]= 3.14

In[35]:= Precision[a]

Out[35]= MachinePrecision
```

通常のパソコンなら機械精度は有効桁数 16 桁となる．3 桁しか入力していないのに，精度が 16 桁になった理由は，入力値を 3.140000000000000 と解釈したためである．

MachinePrecision （機械精度）は，使用している機械が一番能率良く計算をするための精度（浮動小数点演算ユニットがサポートしている精度）であり，計算は非常に速い．しかし，精度が固定されているので，誤差の蓄積および拡大が起こり精度が落ちているような数も，機械精度の数として扱われる．これは厳密な誤差評価を必要とするときには問題となる．そのことを見てみよう．

まず $\frac{1}{3}$ を機械精度に直して， a に代入する．N は浮動小数点数に直すための関数である．

```
In[36]:= a = N[1 / 3]

Out[36]= 0.333333
```

表示されたのは小数点以下 6 桁だけだが，これは，通常の科学計算の結果をざっと見るには 6 桁くらいがちょうど良いからである．次のようにして，その精度を見てみると，MachinePrecision 桁となっている．

```
In[37]:= Precision[a]

Out[37]= MachinePrecision
```

InputForm を用いると，その値が全桁で表示される．

In[38]:= **InputForm[a]**

Out[38]//InputForm=
0.3333333333333333

ここで 10000000000000 a − 3333333333333 を計算して，その結果と精度を見てみよう．この計算の正解は $\frac{1}{3} = 0.33333\cdots$ である．

In[39]:= **b = 10000000000000 a − 3333333333333**

Out[39]= 0.333008

In[40]:= **Precision[b]**

Out[40]= MachinePrecision

In[41]:= **InputForm[b]**

Out[41]//InputForm=
0.3330078125

b の精度は MachinePrecision（機械精度）すなわち 16 桁であると表示されるが，実際にはそれだけの精度を保持していない．

このように，機械精度の実数の計算結果が実際にはどのような精度を保持しているかは，Mathematicaは関知していない．したがって，機械精度で計算を行うときには，その精度について使用者が注意を払う必要がある．

■ 任意精度

機械精度は（16桁に）精度が固定されたものであったが， *Mathematica* ではその他の有限の精度として，任意精度と呼ばれるものがある．

機械精度を超える桁数をもつ実数を入力すると，それは任意精度の数値として扱われる．次の場合は約 18 桁の精度と見なされている．

In[42]:= **a = 1.234567890123456789**

Out[42]= 1.23456789012345679

In[43]:= **Precision[a]**

Out[43]= 18.0915

機械精度よりも低い任意精度の数値を入力することもできる．そのためには数の後にバッククオート \` に続いて有効桁数をつけて表す．

In[44]:= **c = 1.23\`3**

Out[44]= 1.23

In[45]:= **Precision[pi3]**

Out[45]= ∞

関数 N を用いても，任意精度の数値を作れる．

In[46]:= **pi10 = N[Pi, 10]**

Out[46]= 3.141592654

In[47]:= **Precision[pi10]**

Out[47]= 10.

In[48]:= **pi100 = N[Pi, 100]**

Out[48]= 3.141592653589793238462643383279502884197169399375105820974944592307816406286208\
998628034825342117068

In[49]:= **Precision[pi100]**

Out[49]= 100.

いくつかの数値に対して演算をすると，その結果は，一番低い数値の精度となる.

In[50]:= **Precision[Pi + pi10 + pi100]**

Out[50]= 10.4771

Mathematica の数学関数は，引き数の精度に応じて可能な限り高い精度の値を返す.

In[51]:= **Sin[Pi / 3]**

Out[51]= $\frac{\sqrt{3}}{2}$

In[52]:= **Precision[%]**

Out[52]= ∞

In[53]:= **Sin[pi10 / 3]**

Out[53]= 0.8660254038

In[54]:= **Precision[%]**

Out[54]= 10.2185

In[55]:= **Sin[pi100 / 3]**

Out[55]= 0.866025403784438646763723170752936183471402626905190314027903489725966508454400\
018540573093378624287 8

In[56]:= **Precision[%]**

Out[56]= 100.219

■ 演習問題

[3-1]　100 個の異なるものから 30 個を取り出すときの，組み合わせは何通りあるか.

[3-2]　$(3-4\,i)^5$ を展開せよ.

[3-3]　1.2345678901234567890１ という有効数字 20 桁の数に対して，小数部分の逆数をとる，という操作を繰り返し行う．このとき，何回この操作を行うと，有効桁数が 1 桁になるか．

ヒント：小数部分を返す関数は FractionalPart である．

[3-4]　$e^{\pi\sqrt{163}}$ と，その値に一番近い整数との差は，およそどのくらいか．

■ この章で出てきた関数

```
Head[expr]                 expr の頭部を返す
Precision[expr]            expr の精度を返す
N[expr]                    expr の近似値を返す
```